Math 2280-2
Problem 5.4.31

This is a more complicated problem with multiple eigenvalues and chains.

```
> with(LinearAlgebra):
> A:=Matrix(4,4,[35,-12,4,30,  22,-8,3,19,  -10,3,0,-9, -27,9,-3,
  -23]);
```

$$A := \begin{bmatrix} 35 & -12 & 4 & 30 \\ 22 & -8 & 3 & 19 \\ -10 & 3 & 0 & -9 \\ -27 & 9 & -3 & -23 \end{bmatrix} \quad (1)$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} 1 \\ 1 \\ 1 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 & 0 & 0 & 0 \\ -\frac{1}{3} & \frac{1}{3} & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix} \quad (2)$$

Eigenvalue lambda=1 has algebraic multiplicity 4 and we only have two eigenvectors, so this is defective matrix.

We now look at the generalized eigenspace for lambda=1:

```
> Id:=IdentityMatrix(4):
> NullSpace((A-1*Id)^4);
```

$$\left\{ \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix} \right\} \quad (3)$$

The two eigenvectors are in this subspace. Let us pick one of these generalized eigenvectors to do a chain

```
> u1:=Vector([0,1,0,0]);
```

$$u1 := \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \end{bmatrix} \quad (4)$$

```
> u2:= (A-Id) . u1;
```

$$u2 := \begin{bmatrix} -12 \\ -9 \\ 3 \\ 9 \end{bmatrix} \quad (5)$$

```
> u3:= (A-Id) . u2;
```

$$u3 := \begin{bmatrix} -18 \\ -3 \\ 9 \\ 18 \end{bmatrix} \quad (6)$$

```
> u4:=(A-Id) . u3;
```

$$u4 := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (7)$$

This means u3 is an eigenvector. So we have a chain of length 3. (If the chain had stopped with two vectors only, we would have tried another basis vector to get the other chain). We have several linearly independent solutions
One that comes from one of teh eigenvectors

```
> x4:=t-> exp(t)*Vector([0,1/3,1,0]);
```

$$x4 := t \to e^t \, Vector\left(\left[0, \frac{1}{3}, 1, 0\right]\right) \quad (8)$$

And the ones that come from the chain. For convenience I defined some vi's which are the ui's in reverse order (look at algorithm p341)

```
> v1:=u3: v2:=u2: v3:=u1:
> x1:=t-> exp(t)*v1;
  x2:=t-> exp(t)*(t*v1 + v2);
  x3:=t-> exp(t)*(t^2*v1/2 + t*v2 + v3);
```

$$x1 := t \to e^t \, v1$$

$$x2 := t \to e^t \, (t\, v1 + v2)$$

$$x3 := t \to e^t \left(\frac{1}{2} t^2 \, v1 + t\, v2 + v3\right) \quad (9)$$

Let us check that putting these together gives a fundamental solution matrix:

```
> Phi := t -> Matrix(4,4,[x1(t),x2(t),x3(t),x4(t)]);
```

$$\Phi := t \to Matrix(4, 4, [x1(t), x2(t), x3(t), x4(t)]) \quad (10)$$

```
> Phi(t); # ugly expression
```

$$\begin{bmatrix} -18\, e^t & e^t(-18\,t - 12) & e^t\left(-9\,t^2 - 12\,t\right) & 0 \\ -3\, e^t & e^t(-3\,t - 9) & e^t\left(-\frac{3}{2} t^2 - 9\,t + 1\right) & \frac{1}{3} e^t \\ 9\, e^t & e^t(9\,t + 3) & e^t\left(\frac{9}{2} t^2 + 3\,t\right) & e^t \\ 18\, e^t & e^t(18\,t + 9) & e^t\left(9\,t^2 + 9\,t\right) & 0 \end{bmatrix} \quad (11)$$

The following checks that our solution is correct (we should be getting all zeros and the fundamental solution should have linearly indep vectors!)

```
> simplify(map(diff,Phi(t),t) - A . Phi(t));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(12)}$$

```
> Determinant(Phi(0));
```

$$54 \quad \textbf{(13)}$$

```
>
```